# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=python-cppy
pkgver=1.2.1
_commit=13a67972ca9063951440c911270f204aa7fe631a
pkgrel=6
pkgdesc="A collection of C++ headers which make it easier to write Python C extension modules"
url="https://github.com/nucleic/cppy"
license=('BSD')
arch=('any')
# cppy/__init__.py uses setuptools
depends=('python-setuptools')
makedepends=('git' 'python-build' 'python-installer' 'python-setuptools-scm' 'python-wheel')
checkdepends=('python-pytest')
source=("git+https://github.com/nucleic/cppy.git#commit=$_commit")
sha512sums=('SKIP')

build() {
  cd cppy
  python -m build -wn
}

check() {
  cd cppy
  python -m pytest
}

package() {
  cd cppy
  python -m installer -d "$pkgdir" dist/*.whl
  install -Dm644 LICENSE -t "$pkgdir"/usr/share/licenses/$pkgname/
}
